Computer
Organization and Structure

Bing-Yu Chen
National Taiwan University
Instructions: Language of the Computer

- Operations and Operands
  - of the Computer Hardware
- Signed and Unsigned Numbers
- Representing Instructions
  - in the Computer
- Logical Operations
- Instructions for Making Decisions
- Supporting Procedures
  - in Computer Hardware
- Communicating with People
- MIPS Addressing
  - for 32-Bit Immediates and Addresses
- Translating and Starting a Program
- Arrays vs. Pointers
Instruction Set

- The repertoire of instructions of a computer
- Different computers have different instruction sets
  - But with many aspects in common
- Early computers had very simple instruction sets
  - Simplified implementation
- Many modern computers also have simple instruction sets
The MIPS Instruction Set

- Used as the example throughout the book
- Stanford MIPS commercialized by MIPS Technologies (www.mips.com)
- Large share of embedded core market
  - Applications in consumer electronics, network/storage equipment, cameras, printers, ...
- Typical of many modern ISAs
  - See MIPS Reference Data tear-out card, and Appendixes B and E
Arithmetic Operations

- Add and Subtract, 3 operands
  - 2 sources and 1 destination
- Operand order is fixed
  - Destination first
  - All arithmetic operations have this form

Example:
- C code: \( a = b + c \)
- MIPS code: `add a, b, c`
Arithmetic Operations

- **Design Principle 1:**
  - Simplicity favors regularity
    - Regularity makes implementation simpler
    - Simplicity enables higher performance at lower cost
Arithmetic Examples

- compiling two C assignments into MIPS
  - C code:  
    a = b + c;
    d = a - e;
  - MIPS code:  
    add a, b, c
    sub d, a, e

- compiling a complex C assignment into MIPS
  - C code:  
    f = (g + h) - (i + j)
  - MIPS code:  
    add $t0, g, h  # temp t0 = g + h
    add $t1, i, j  # temp t1 = i + j
    sub f, $t0, $t1  # f = t0 - t1
Register Operands

- Of course this complicates some things...
  - C code: \( a = b + c + d; \)
  - MIPS code: `add a, b, c`  
    `add a, a, d`
  - where \( a & b & c & d \) mean **registers**

- Arithmetic instructions use register operands
  - operands must be **registers**
Register Operands

- MIPS has a $32 \times 32$-bit register file
  - Use for frequently accessed data
  - Numbered 0 to 31
  - 32-bit data called a “word”

- Assembler names
  - $t0, t1, \ldots, t9$ for temporary values
  - $s0, s1, \ldots, s7$ for saved variables

- Design Principle 2:
  - smaller is faster
    - c.f. main memory: millions of locations
Register Operand Example

- C code: \( f = (g + h) - (i + j) \)
  - assume \( f, ..., j \) in \( s0, ..., s4 \)

- MIPS code: add \( t0, s1, s2 \)
  add \( t1, s3, s4 \)
  sub \( s0, t0, t1 \)
Registers vs. Memory

- Arithmetic instructions operands must be registers
  - only 32 registers provided
- Compiler associates variables with registers
- What about programs with lots of variables

```
Control
Datapath
Memory
Input
Output
```

processor

I/O
Memory Operands

- Main memory used for composite data
  - Arrays, structures, dynamic data
- To apply arithmetic operations
  - Load values from memory into registers
  - Store result from register to memory
- MIPS is Big Endian
  - Most-Significant Byte at least address of a word
  - c.f. Little Endian: Least-Significant Byte at least address
Memory Organization

- viewed as a large, single-dimension array, with an address
- A memory address is an index into the array
- "byte addressing" means that the index points to a 8-bit byte of memory
Big Endian vs. Little Endian

Big Endianness

Little Endianness

Memory
Memory Organization

- "bytes" are nice, but most data items use larger "words"
- for MIPS
  - a word is 32 bits or 4 bytes
- $2^{32}$ bytes with byte addresses from 0 to $2^{32}-1$
- $2^{30}$ words with byte addresses 0, 4, 8, ... $2^{32}-4$
- words are aligned (*alignment restriction*)
  - Address must be a multiple of 4
  - What are the least 2 significant bits of a word address?
Word Addressing

- Address
  - Index
  - Byte offset
  - 32 bits
  - Data

- 1G entries
- 32 bits of data
- Data

- 0
  - 32 bits of data
- 4
  - 32 bits of data
- 8
  - 32 bits of data
- 12
  - 32 bits of data
- ...
Load & Store Instructions

- C code: \( g = h + A[8] \);
  - \( g \) in \$s1, \( h \) in \$s2, \textit{base address} of \( A \) in \$s3

- MIPS code: \( \text{lw } \$t0, 32(\$s3) \)
  - \( \text{add } \$s1, \$s2, \$t0 \)
  - index 8 requires \textit{offset} of 32
  - 4 bytes per word

- can refer to registers by name (e.g., \$s2, \$t0) instead of number
Load & Store Instructions

  - h in $s2, base address of A in $s3

- MIPS code: \[
\begin{align*}
  \text{lw} & \quad $t0, 32($s3) \\
  \text{add} & \quad $t0, $s2, $t0 \\
  \text{sw} & \quad $t0, 48($s3)
\end{align*}
\]

- store word has *destination last*

- remember arithmetic operands are registers, not memory
  - can’t write: \[ \text{add} 48($s3), $s2, 32($s3) \]
Registers vs. Memory

- Registers are faster to access than memory
- Operating on memory data requires loads and stores
  - More instructions to be executed
- Compiler must use registers for variables as much as possible
  - Only spill to memory for less frequently used variables
  - Register optimization is important!
Constants

- Small constants are used quite frequently
  - e.g., \( A = A + 5; \)
  - \( B = B + 1; \)
  - \( C = C - 18; \)

- Solutions? Why not?
  - Put 'typical constants' in memory and load them?
  - Create hard-wired registers (like $zero) for constants like one?
The Constant Zero

- MIPS register 0 ($zero) is the constant 0
  - Cannot be overwritten
- Useful for common operations
  - add $t2, $s1, $zero
    - e.g., move between registers
Immediate Operands

- Constant data specified in an instruction
  - addi $s3, $s3, 4

- No subtract immediate* instruction
  - Just use a negative constant
  - addi $s2, $s1, -1

- Design Principle 3:
  - Make the common case fast
    - Small constants are common
    - Immediate operand avoids a load instruction

* e.g. subi
Numbers

- Bits are just bits (no inherent meaning)
  - conventions define relationship between bits and numbers

- Binary numbers (base 2)
  - decimal: $0...2^n-1$

- Of course it gets more complicated:
  - numbers are finite (overflow)
  - fractions and real numbers
  - negative numbers
Unsigned Binary Integers

- Given an n-bit number

\[ x = x_{n-1}2^{n-1} + x_{n-2}2^{n-2} + \ldots + x_12^1 + x_02^0 \]

- Range: 0 to \(+2^n - 1\)

- Example

  - 0000 0000 0000 0000 0000 0000 0000 1011_2
    
    \[
    = 0 + \ldots + 1 \times 2^3 + 0 \times 2^2 + 1 \times 2^1 + 1 \times 2^0
    \]
    
    \[
    = 0 + \ldots + 8 + 0 + 2 + 1 = 11_{10}
    \]

- Using 32 bits

  - 0 to \(+4,294,967,295\)
2’s-Complement Signed Integers

- Given an n-bit number
  \[ x = -x_{n-1}2^{n-1} + x_{n-2}2^{n-2} + \ldots + x_12^1 + x_02^0 \]

- Range: \(-2^{n-1}\) to \(+2^{n-1} - 1\)

- Example
  - \(1111\ 1111\ 1111\ 1111\ 1111\ 1111\ 1111\ 1100_2\)
  - \(-1 \times 2^{31} + 1 \times 2^{30} + \ldots + 1 \times 2^2 + 0 \times 2^1 + 0 \times 2^0\)
  - \(-2,147,483,648 + 2,147,483,644 = -4_{10}\)

- Using 32 bits
  - \(-2,147,483,648\) to \(+2,147,483,647\)
2’s-Complement Signed Integers

- Bit 31 is sign bit
  - 1 for negative numbers
  - 0 for non-negative numbers
- $-(−2^n - 1)$ can’t be represented
- Non-negative numbers have the same unsigned and 2’s-complement representation
- Some specific numbers
  - 0: 0000 0000 ... 0000
  - −1: 1111 1111 ... 1111
  - Most-negative: 1000 0000 ... 0000
  - Most-positive: 0111 1111 ... 1111
Signed Negation

- Complement and add 1
  - Complement means $1 \rightarrow 0$, $0 \rightarrow 1$
    \[ x + \overline{x} = 1111\ldots111_2 = -1 \]
    \[ \overline{x} + 1 = -x \]

- Example: negate +2
  - $+2 = 0000\ 0000\ \ldots\ 0010_2$
  - $-2 = 1111\ 1111\ \ldots\ 1101_2 + 1$
    \[ = 1111\ 1111\ \ldots\ 1110_2 \]

- “negate” and “complement” are quite different!
Sign Extension

- Representing a number using more bits
  - Preserve the numeric value
- In MIPS instruction set
  - addi: extend immediate value
  - lb, lh: extend loaded byte/halfword
  - beq, bne: extend the displacement
- Replicate the sign bit to the left
  - c.f. unsigned values: extend with 0s
- Examples: 8-bit to 16-bit
  - +2: 0000 0010 => 0000 0000 0000 0010
  - -2: 1111 1110 => 1111 1111 1111 1110
Representing Instructions

- Instructions are encoded in binary
  - Called **machine code**

- MIPS instructions
  - Encoded as **32-bit** instruction words
  - Small number of formats encoding operation code (opcode), register numbers, ...
  - Regularity!

- Register numbers
  - \$t0 – \$t7 are reg’s 8 – 15
  - \$t8 – \$t9 are reg’s 24 – 25
  - \$s0 – \$s7 are reg’s 16 – 23
MIPS R-format Instructions

- **op** = operation code (opcode)
  - basic operation of the instruction
- **rs / rt / rd**
  - register source / destination operand
- **shamt = shift amount**
  - 000000 for now
- **funct = function code**
  - extends opcode

<table>
<thead>
<tr>
<th>op</th>
<th>rs</th>
<th>rt</th>
<th>rd</th>
<th>shamt</th>
<th>funct</th>
</tr>
</thead>
<tbody>
<tr>
<td>6 bits</td>
<td>5 bits</td>
<td>5 bits</td>
<td>5 bits</td>
<td>5 bits</td>
<td>6 bits</td>
</tr>
</tbody>
</table>
R-format Example

- add $t0, $s1, $s2

<table>
<thead>
<tr>
<th>special</th>
<th>$s1</th>
<th>$s2</th>
<th>$t0</th>
<th>0</th>
<th>add</th>
</tr>
</thead>
<tbody>
<tr>
<td>0</td>
<td>17</td>
<td>18</td>
<td>8</td>
<td>0</td>
<td>32</td>
</tr>
<tr>
<td>000000</td>
<td>10001</td>
<td>10010</td>
<td>01000</td>
<td>00000</td>
<td>100000</td>
</tr>
</tbody>
</table>

- \[ \text{0000001000110010010000000001000000}_2 = 02324020_{16} \]
Hexadecimal

- **Base 16**
  - Compact representation of bit strings
  - 4 bits per hex digit

<p>| | | | | | | | |</p>
<table>
<thead>
<tr>
<th></th>
<th></th>
<th></th>
<th></th>
<th></th>
<th></th>
<th></th>
<th></th>
</tr>
</thead>
<tbody>
<tr>
<td>0</td>
<td>0000</td>
<td>4</td>
<td>0100</td>
<td>8</td>
<td>1000</td>
<td>c</td>
<td>1100</td>
</tr>
<tr>
<td>1</td>
<td>0001</td>
<td>5</td>
<td>0101</td>
<td>9</td>
<td>1001</td>
<td>d</td>
<td>1101</td>
</tr>
<tr>
<td>2</td>
<td>0010</td>
<td>6</td>
<td>0110</td>
<td>a</td>
<td>1010</td>
<td>e</td>
<td>1110</td>
</tr>
<tr>
<td>3</td>
<td>0011</td>
<td>7</td>
<td>0111</td>
<td>b</td>
<td>1011</td>
<td>f</td>
<td>1111</td>
</tr>
</tbody>
</table>

- **Example: eca8 6420**
  - 1110 1100 1010 1000 0110 0100 0010 0000
MIPS I-format Instructions

<table>
<thead>
<tr>
<th>op</th>
<th>rs</th>
<th>rt</th>
<th>constant or address</th>
</tr>
</thead>
<tbody>
<tr>
<td>6 bits</td>
<td>5 bits</td>
<td>5 bits</td>
<td>16 bits</td>
</tr>
</tbody>
</table>

- Immediate arithmetic and load/store instructions
  - rs / rt: source or destination register number
  - Constant: \(-2^{15}\) to \(+2^{15} - 1\)
  - Address: offset added to base address in rs

- **Design Principle 4:**
  - Good design demands good compromises
    - Different formats complicate decoding, but allow 32-bit instructions uniformly
    - Keep formats as similar as possible
I-format Example

<table>
<thead>
<tr>
<th></th>
<th>op</th>
<th>rs</th>
<th>rt</th>
<th>constant or address</th>
</tr>
</thead>
<tbody>
<tr>
<td>6 bits</td>
<td>5 bits</td>
<td>5 bits</td>
<td>16 bits</td>
<td></td>
</tr>
</tbody>
</table>

 lw $t0, 32($s2)

<table>
<thead>
<tr>
<th>lw</th>
<th>$s2</th>
<th>$t0</th>
<th>32</th>
</tr>
</thead>
<tbody>
<tr>
<td>35</td>
<td>18</td>
<td>8</td>
<td>32</td>
</tr>
<tr>
<td>100011</td>
<td>10010</td>
<td>01000</td>
<td>000000000001000000</td>
</tr>
</tbody>
</table>
C / MIPS / Machine Languages

- C: \[ A[300] = h + A[300] \]
- MIPS: \( \text{lw} \$t0, 1200(\$t1) \)
  \( \text{add} \$t0, \$s2, \$t0 \)
  \( \text{sw} \$t0, 1200(\$t1) \)

- Machine Language:

<table>
<thead>
<tr>
<th>35</th>
<th>9</th>
<th>8</th>
<th>1200</th>
</tr>
</thead>
<tbody>
<tr>
<td>0</td>
<td>18</td>
<td>8</td>
<td>8</td>
</tr>
<tr>
<td>43</td>
<td>9</td>
<td>8</td>
<td>1200</td>
</tr>
</tbody>
</table>
Stored Program Concept

- Instructions are bits
- Programs are stored in memory
  - to be read or written just like data
- Fetch & Execute Cycle
  - Instructions are fetched and put into a special register
  - Bits in the register "control" the subsequent actions
  - Fetch the “next” instruction and continue

Memory

- Accounting program (machine code)
- Editor program (machine code)
- C compiler (machine code)
- Payroll data
- Book text
- C code for editor program

memory for data, programs, compilers, editors, etc.
## Logical Operations

<table>
<thead>
<tr>
<th>Logical operations</th>
<th>C operators</th>
<th>MIPS instructions</th>
</tr>
</thead>
<tbody>
<tr>
<td>Shift left</td>
<td><code>&lt;&lt;</code></td>
<td><code>sll</code></td>
</tr>
<tr>
<td>Shift right</td>
<td><code>&gt;&gt;</code></td>
<td><code>srl</code></td>
</tr>
<tr>
<td>Bitwise AND</td>
<td><code>&amp;</code></td>
<td><code>and, andi</code></td>
</tr>
<tr>
<td>Bitwise OR</td>
<td>`</td>
<td>`</td>
</tr>
</tbody>
</table>
Logical Operations

<table>
<thead>
<tr>
<th>Instruction</th>
<th>Example</th>
</tr>
</thead>
<tbody>
<tr>
<td>and</td>
<td>and $s1, $s2, $s3</td>
</tr>
<tr>
<td>or</td>
<td>or $s1, $s2, $s3</td>
</tr>
<tr>
<td>nor</td>
<td>nor $s1, $s2, $s3</td>
</tr>
<tr>
<td>and immediate</td>
<td>andi $s1, $s2, 100</td>
</tr>
<tr>
<td>or immediate</td>
<td>ori $s1, $s2, 100</td>
</tr>
<tr>
<td>shift left logical</td>
<td>sll $s1, $s2, 10</td>
</tr>
<tr>
<td>shift right logical</td>
<td>srl $s1, $s2, 10</td>
</tr>
</tbody>
</table>

There is no NOT, since ...

- NOT (A) = NOT (A OR 0) = A NOR 0
Shift Operations

- **shamt**: how many positions to shift
- **Shift left logical**
  - Shift left and fill with 0 bits
  - sll by \(i\) bits multiplies by \(2^i\)
- **Shift right logical**
  - Shift right and fill with 0 bits
  - srl by \(i\) bits divides by \(2^i\) (unsigned only)

<table>
<thead>
<tr>
<th>op</th>
<th>rs</th>
<th>rt</th>
<th>rd</th>
<th>shamt</th>
<th>funct</th>
</tr>
</thead>
<tbody>
<tr>
<td>6 bits</td>
<td>5 bits</td>
<td>5 bits</td>
<td>5 bits</td>
<td>5 bits</td>
<td>6 bits</td>
</tr>
</tbody>
</table>
Shift Operations

- **NOTICE**
  - shift left/right logical is not *I-type*

- **Example**: `sll $t2, $s0, 4`

- **Machine Language**:

<table>
<thead>
<tr>
<th>op</th>
<th>rs</th>
<th>rt</th>
<th>rd</th>
<th>shamt</th>
<th>funct</th>
</tr>
</thead>
<tbody>
<tr>
<td>special</td>
<td>none</td>
<td>$s0</td>
<td>$t2</td>
<td>4</td>
<td>sll</td>
</tr>
<tr>
<td>0</td>
<td>0</td>
<td>16</td>
<td>10</td>
<td>4</td>
<td>0</td>
</tr>
</tbody>
</table>
AND Operations

- Useful to mask bits in a word
  - Select some bits, clear others to 0

- and $t0, $t1, $t2
  - $t2 = 0000 0000 0000 0000 0000 1101 1100 0000
  - $t1 = 0000 0000 0000 0000 0011 1100 0000 0000
  - $t0 = 0000 0000 0000 0000 0000 1100 0000 0000
OR Operations

- Useful to include bits in a word
  - Set some bits to 1, leave others unchanged

- or $t0, $t1, $t2
  - $t2 = 0000 0000 0000 0000 0000 1101 1100 0000
  - $t1 = 0000 0000 0000 0000 0011 1100 0000 0000
  - $t0 = 0000 0000 0000 0000 0011 1101 1100 0000
NOT Operations

- Useful to invert bits in a word
  - Change 0 to 1, and 1 to 0

- MIPS has NOR 3-operand instruction
  - \( a \) NOR \( b \) == NOT ( \( a \) OR \( b \) )

- nor \$t0, \$t1, \$zero
  - \$t1 = 0000 0000 0000 0000 0011 1100 0000 0000
  - \$t0 = 1111 1111 1111 1111 1100 0011 1111 1111
Conditional Operations

- Decision making instructions
  - alter the control flow,
  - i.e., change the "next" instruction to be executed

- MIPS **conditional branch** instructions:
  - `bne $t0, $t1, Label`
  - `beq $t0, $t1, Label`

Example:

```plaintext
if (i==j) h = i + j;

bne $s0, $s1, Label
add $s3, $s0, $s1
Label: ....
```
Unconditional Operations

- MIPS unconditional branch instructions:
  - `j Label`

- (Un-)Conditional Branch Example:
  ```
  if (i==j)    bne $s3, $s4, Else
  f=g+h;      add $s0, $s1, $s2
  else        j Exit
   f=g-h;     Else: sub $s0, $s1, $s2
   Exit:      ...          
  ```

- Can you build a simple `for / while` loop?
While Loop

C:
while (save [i] == k) i += 1;
■ assume i in $s3, k in $s5, address of save in $s6

MIPS:
Loop: sll $t1, $s3, 2 # $t1=4*i
     add $t1, $t1, $s6 # $t1=addr. of save[i]
     lw $t0, 0($t1) # $t0=save[i]
     bne $t0, $s5, Exit # go to Exit if save[i]!=k
     addi $s3, $s3, 1 # i+=1
     j Loop # go to Loop

Exit:
Control Flow

- set on less than:
  ```
  if ($s3 < $s4)   slt  $t1, $s3, $s4
      $t1=1;
  else
      $t1=0;
  ```

- can use this instruction to build
  ```
  "blt $s1, $s2, Label"
  ```
  can now build general control structures

- NOTE
  - the assembler needs a register to do this,
  - there are policy of use conventions for registers
Branch Instruction Design

- Why not blt, bge, etc?
- Hardware for $<$, $\geq$, ... slower than $=$, $\neq$
  - Combining with branch involves more work per instruction, requiring a slower clock
  - All instructions penalized!
- beq and bne are the common case
- This is a good design compromise
Case/Switch in C

```c
switch (k) {
    case 0:    f=i+j;         break;    /* k=0 */
    case 1:    f=g+h;         break;    /* k=1 */
    case 2:    f=g-h;         break;    /* k=2 */
    case 3:    f=i-j;         break;    /* k=3 */
}
```
Case/Switch in MIPS
-- using Jump Address Table

```
slt  $t3, $s5, $zero      # test if k < 0
bne  $t3, $zero, Exit    # if k < 0, go to Exit
slti $t3, $s5, 4         # test if k < 4
beq  $t3, $zero, Exit    # if k >= 4, go to Exit
sll  $t1, $s5, 2         # $t1 = 4 * k
add  $t1, $t1, $t4       # $t1=addr. of JumpTable[k]
lw   $t0, 0($t1)         # $t0= JumpTable[k]
jr   $t0                 # jump based on $t0
```

```
L0:  add  $s0, $s3, $s4    # k = 0, so f gets i + j
     j Exit              # end of this case so go to Exit
L1:  add  $s0, $s1, $s2    # k = 1, so f gets g + h
     j Exit              # end of this case so go to Exit
L2:  sub  $s0, $s1, $s2    # k = 2, so f gets g - h
     j Exit              # end of this case so go to Exit
L3:  sub  $s0, $s3, $s4    # k = 3, so f gets i - j
     j Exit              # end of switch statement
```
Signed vs. Unsigned

- Signed comparison: slt, slti
- Unsigned comparison: sltu, sltui
- Example
  - $s0 = 1111 1111 1111 1111 1111 1111 1111 1111
  - $s1 = 0000 0000 0000 0000 0000 0000 0000 0001
  - \texttt{slt } \$t0, \$s0, \$s1 \# signed
    - $-1 < +1 \Rightarrow \$t0 = 1$
  - \texttt{sltu } \$t0, \$s0, \$s1 \# unsigned
    - $+4,294,967,295 > +1 \Rightarrow \$t0 = 0$
Procedure Calling

- Steps required
  - Place parameters in registers
  - Transfer control to procedure
  - Acquire storage for procedure
  - Perform procedure’s operations
  - Place result in register for caller
  - Return to place of call
# Register Usage

<table>
<thead>
<tr>
<th>Name</th>
<th>Register No.</th>
<th>Usage</th>
</tr>
</thead>
<tbody>
<tr>
<td>$zero</td>
<td>0</td>
<td>the constant value 0</td>
</tr>
<tr>
<td>$v0-$v1</td>
<td>2-3</td>
<td>values for results &amp; expression evaluation</td>
</tr>
<tr>
<td>$a0-$a3</td>
<td>4-7</td>
<td>arguments</td>
</tr>
<tr>
<td>$t0-$t7</td>
<td>8-15</td>
<td>temporaries (can be overwritten by callee)</td>
</tr>
<tr>
<td>$s0-$s7</td>
<td>16-23</td>
<td>saved (must be saved/restored by callee)</td>
</tr>
<tr>
<td>$t8-$t9</td>
<td>24-25</td>
<td>more temporaries</td>
</tr>
<tr>
<td>$gp</td>
<td>28</td>
<td>global pointer</td>
</tr>
<tr>
<td>$sp</td>
<td>29</td>
<td>stack pointer</td>
</tr>
<tr>
<td>$fp</td>
<td>30</td>
<td>frame pointer</td>
</tr>
<tr>
<td>$ra</td>
<td>31</td>
<td>return address</td>
</tr>
</tbody>
</table>

Register 1 ($at) reserved for assembler, 26-27 for operating system.
Procedure Call Instructions

- Procedure call: jump and link
  - `jal ProcedureLabel`
    - Address of following instruction put in $ra
    - Jumps to target address

- Procedure return: jump register
  - `jr $ra`
    - Copies $ra to program counter
    - Can also be used for computed jumps
      - e.g., for case/switch statements
Leaf Procedure in C

int leaf_example (int g, int h, int i, int j) {
    int f;

    f = (g+h)-(i+j);
    return f;
}

- Assume
  - Arguments g, ..., j in $a0, ..., $a3
  - f in $s0 (hence, need to save $s0 on stack)
  - Result in $v0
Leaf Procedure in MISP

\texttt{addi \$sp, \$sp, -4} \quad \# \text{adjust stack for saving \$s0} \\
\texttt{sw \$s0, 0(\$sp)} \\
\texttt{add \$t0, \$a0, \$a1} \quad \# \text{g+h} \\
\texttt{add \$t1, \$a2, \$a3} \quad \# \text{i+j} \\
\texttt{sub \$s0, \$t0, \$t1} \quad \# (g+h)-(i+j) \\
\texttt{add \$v0, \$s0, \$zero} \quad \# \text{return } f (\$v0=\$s0+0) \\
\texttt{lw \$s0, 0(\$sp)} \\
\texttt{addi \$sp, \$sp, 4} \quad \# \text{adjust stack again} \\
\texttt{jr \$ra} \quad \# \text{jump back to calling routine}
The Stack

High address

Low address

contents of register $s0

$sp

$sp
Non-Leaf Procedures

- Procedures that call other procedures
- For nested call, caller needs to save on the stack:
  - Its return address
  - Any arguments and temporaries needed after the call
- Restore from the stack after the call
Recursive Procedure in C

int fact (int n) {
    if (n < 1)
        return 1;
    else
        return (n * fact (n - 1));
}

☐ Assume
  ● Argument n in $a0
  ● Result in $v0
Recursive Procedure in MISP

fact:
addi $sp, $sp, -8 # adjust stack for 2 items
sw $ra, 4($sp) # save the return address
sw $a0, 0($sp) # save the argument n
slti $t0, $a0, 1 # test for n < 1
beq $t0, $zero, L1 # if n >= 1, go to L1
addi $sp, $sp, 8 # pop 2 items off stack
addi $v0, $zero, 1 # return 1
jr $ra # return to after jal
L1: addi $a0, $a0, -1 # n >= 1: argument gets (n - 1)
jal fact # call fact with (n - 1)
lw $a0, 0($sp) # return from jal: restore argument n
lw $ra, 4($sp) # restore the return address
addi $sp, $sp, 8 # adjust stack pointer to pop 2 items
mul $v0, $a0, $v0 # return n * fact (n - 1)
jr $ra # return to the caller
Local Data on the Stack

Local data allocated by callee
- e.g., C automatic variables
- Procedure frame (activation record)
  - Used by some compilers to manage stack storage

Diagram:
- High address
- $fp$ (frame pointer)
- $sp$ (stack pointer)
- Saved argument registers (if any)
- Saved return address
- Saved saved registers (if any)
- Local arrays and structures (if any)
## Memory Layout

- **Text**: program code
- **Static data**: global variables
  - e.g., static variables in C, constant arrays and strings
  - $gp$ initialized to address allowing $\pm$ offsets into this segment
- **Dynamic data**: heap
  - E.g., malloc in C
- **Stack**: automatic storage

### Memory Map

<table>
<thead>
<tr>
<th>Stack</th>
<th>Dynamic data</th>
<th>Static data</th>
<th>Text</th>
</tr>
</thead>
<tbody>
<tr>
<td>$sp \rightarrow 7ffe fffe_{hex}$</td>
<td>$gp \rightarrow 1000 8000_{hex}$</td>
<td>$gp \rightarrow 1000 0000_{hex}$</td>
<td>$pc \rightarrow 0040 0000_{hex}$</td>
</tr>
</tbody>
</table>
Character Data

- Byte-encoded character sets
  - ASCII: 128 characters
    - 95 graphic, 33 control
  - Latin-1: 256 characters
    - ASCII, +96 more graphic characters
- Unicode: 32-bit character set
  - Used in C++ wide characters, ...
  - Most of the world’s alphabets, plus symbols
  - UTF-8, UTF-16: variable-length encodings
Byte/Halfword Operations

- Could use bitwise operations
- MIPS byte/halfword load/store
  - String processing is a common case

- `lb rt, offset(rs)`    `lh rt, offset(rs)`
  - Sign extend to 32 bits in rt
- `lbu rt, offset(rs)`    `lhu rt, offset(rs)`
  - Zero extend to 32 bits in rt
- `sb rt, offset(rs)`    `sh rt, offset(rs)`
  - Store just rightmost byte/halfword
String Copy Procedure in C

```c
void strcpy (char x[], char y []) {
    int i;

    i = 0;
    while (x[i] != y[i] || y[i] == '\0') {
        i = i + 1;
    }
}
```

- Assume
  - Null-terminated string
  - Addresses of x, y in $a0, $a1, i in $s0
String Copy Procedure in MIPS

```assembly
addi $sp, $sp, -4
sw $s0, 0($sp)
add $s0, $zero, $zero  # i = 0
L1:add $t1, $s0, $a1  # address of y[i] in $t1
lb $t2, 0($t1)  # $t2 = y[i]
add $t3, $s0, $a0  # address of x[i] in $t3
sb $t2, 0($t3)  # x[i] = y[i]
beq $t2, $zero, L2  # if y[i] == 0, go to L2
addi $s0, $s0, 1  # i = i + 1
j L1  # go to L1
L2:lw $s0, 0($sp)  # restore old $s0
add $sp, $sp, 4
jr $ra
```
32-bit Constants

- Most constants are small
  - 16-bit immediate is sufficient
- For the occasional 32-bit constant
  - lui rt, constant
    - Copies 16-bit constant to left 16 bits of rt
    - Clears right 16 bits of rt to 0

\[
\begin{array}{c}
lui \; s0, \; 61 \\
or\; i\; s0, \; s0, \; 2304
\end{array}
\]

\[
\begin{array}{c|c}
0000 & 0000 & 0111 & 1101 & 0000 & 0000 & 0000 & 0000 \\
0000 & 0000 & 0111 & 1101 & 0000 & 1001 & 0000 & 0000
\end{array}
\]
Branch Addressing

- **Instructions:**
  - `bne $s0,$s1,L1`
  - `beq $s0,$s1,L2`

- **Formats:**

<table>
<thead>
<tr>
<th>I</th>
<th>op</th>
<th>rs</th>
<th>rt</th>
<th>16 bit number</th>
</tr>
</thead>
</table>

- Most branch targets are near branch
  - Forward or backward
- **PC-relative addressing**
  - Target address = PC + offset × 4
  - PC already incremented by 4 by this time
Jump Addressing

Instructions:
- `j L1`
- `jal L2`

Formats:
- Jump targets could be anywhere in text segment
  - Encode full address in instruction
- (Pseudo)Direct jump addressing
  - Target address = $\text{PC}_{31...28}$ : $(\text{address} \times 4)$
Target Addressing Example

C:
while (save [i] == k) i += 1;

MIPS:

<table>
<thead>
<tr>
<th>Opcode</th>
<th>Instruction</th>
<th>Address</th>
<th>0</th>
<th>19</th>
<th>9</th>
<th>4</th>
<th>0</th>
</tr>
</thead>
<tbody>
<tr>
<td>sll</td>
<td>$t1, $s3, 2</td>
<td>80000</td>
<td>0</td>
<td></td>
<td></td>
<td></td>
<td></td>
</tr>
<tr>
<td>add</td>
<td>$t1, $t1, $s6</td>
<td>80004</td>
<td>0</td>
<td>9</td>
<td>22</td>
<td>9</td>
<td>0</td>
</tr>
<tr>
<td>lw</td>
<td>$t0, 0($t1)</td>
<td>80008</td>
<td>35</td>
<td>9</td>
<td>8</td>
<td></td>
<td>0</td>
</tr>
<tr>
<td>bne</td>
<td>$t0, $s5, Exit</td>
<td>80012</td>
<td>5</td>
<td>8</td>
<td>21</td>
<td></td>
<td>2</td>
</tr>
<tr>
<td>addi</td>
<td>$s3, $s3, 1</td>
<td>80016</td>
<td>8</td>
<td>19</td>
<td>19</td>
<td></td>
<td>1</td>
</tr>
<tr>
<td>j</td>
<td>Loop</td>
<td>80020</td>
<td>2</td>
<td></td>
<td></td>
<td></td>
<td>20000</td>
</tr>
</tbody>
</table>

Exit: 80024 ...
Branching Far Away

- If branch target is too far to encode with 16-bit offset, assembler rewrites the code

Example

```assembly
beq $s0,$s1, L1
↓
bne $s0,$s1, L2
j L1
L2:     ...
```
Addressing Modes

- Immediate addressing

- Register addressing
Addressing Modes

- Base addressing

<table>
<thead>
<tr>
<th>op</th>
<th>rs</th>
<th>rt</th>
<th>address</th>
</tr>
</thead>
</table>

Register

- Memory
  - Byte
  - Halfword
  - Word
Addressing Modes

- PC-relative addressing

- Pseudodirect addressing
Decoding Machine Code

What is the assembly language statement corresponding to this machine instruction?

- 00af8020_{\text{hex}}
  - 0000 0000 1010 1111 1000 0000 0010 0000
- \text{op} = 000000 \Rightarrow \text{R-format}
- \text{rs} = 00101 (a1)/ \text{rt} = 01111 (t7)/ \text{rd} = 10000 (s0)
- \text{shamt} = 00000 / \text{funct} = 100000 \Rightarrow \text{add}

- add $s0, $a1, $t7
Translation and Startup

C program

Assembly language program

Many compilers produce object modules directly

Object: Machine language module

Object: Library routine

Static linking

Machine language program

Linker

Loader

Memory
Assembler Pseudoinstructions

- Most assembler instructions represent machine instructions one-to-one
- Pseudoinstructions: figments of the assembler’s imagination
  - `move $t0, $t1` → `add $t0, $zero, $t1`
  - `blt $t0, $t1, L` → `slt $at, $t0, $t1`
    - `bne $at, $zero, L`
  - `$at` (register 1): assembler temporary
Producing an Object Module

- Assembler (or compiler) translates program into machine instructions
- Provides information for building a complete program from the pieces
  - Header: described contents of object module
  - Text segment: translated instructions
  - Static data segment: data allocated for the life of the program
  - Relocation info: for contents that depend on absolute location of loaded program
  - Symbol table: global definitions and external refs
  - Debug info: for associating with source code
Linking Object Modules

- Produces an executable image
  1. Merges segments
  2. Resolve labels (determine their addresses)
  3. Patch location-dependent and external refs

- Could leave location dependencies for fixing by a relocating loader
  - But with virtual memory, no need to do this
  - Program can be loaded into absolute location in virtual memory space
Loading a Program

- Load from image file on disk into memory
  1. Read header to determine segment sizes
  2. Create virtual address space
  3. Copy text and initialized data into memory
     - Or set page table entries so they can be faulted in
  4. Set up arguments on stack
  5. Initialize registers (including $sp, $fp, $gp)
  6. Jump to startup routine
     - Copies arguments to $a0, ... and calls main
     - When main returns, do exit syscall
Dynamic Linking

- Only link/load library procedure when it is called
  - Requires procedure code to be relocatable
  - Avoids image bloat caused by static linking of all (transitively) referenced libraries
  - Automatically picks up new library versions
Lazy Linkage

Indirection table

Stub: Loads routine ID, Jump to linker/loader

Linker/loader code

Dynamically mapped code

First call to DLL routine

Subsequent calls to DLL routine
Arrays vs. Pointers

- Array indexing involves
  - Multiplying index by element size
  - Adding to array base address

- Pointers correspond directly to memory addresses
  - Can avoid indexing complexity
Array vs. Pointers in C

```c
void clear1 (int array[], int size) {
    int i;
    for (i = 0; i < size; i += 1)
        array[i] = 0;
}

void clear2 (int *array, int size) {
    int *p;
    for (p = &array[0]; p < &array[size]; p += 1)
        *p = 0;
}
```
Array Version of Clear in MIPS

```
add $t0, $zero, $zero
loop1: sll $t1, $t0, 2
add $t2, $a0, $t1
sw $zero, 0($t2)
addi $t0, $t0, 1
slt $t3, $t0, $a1
bne $t3, $zero, loop1
```
Pointer Version of Clear in MIPS

```
add       $t0, $a0, $zero
loop2:    sw       $zero, 0($t0)
addi      $t0, $t0, 4
sll       $t1, $a1, 2
add       $t2, $a0, $t1
slt       $t3, $t0, $t2
bne       $t3, $zero, loop2
```
New Pointer Version of Clear

add $t0, $a0, $zero
sll $t1, $a1, 2
add $t2, $a0, $t1

loop2:
    sw $zero, 0($t0)
addi $t0, $t0, 4
slt $t3, $t0, $t2
bne $t3, $zero, loop2
Comparing the Two Versions

<p>| | | |</p>
<table>
<thead>
<tr>
<th></th>
<th></th>
<th></th>
</tr>
</thead>
<tbody>
<tr>
<td>add</td>
<td>$t0, $zero, $zero</td>
<td>add</td>
</tr>
<tr>
<td>lp1:</td>
<td>sll</td>
<td>$t1, $t0, 2</td>
</tr>
<tr>
<td></td>
<td>add</td>
<td>$t2, $a0, $t1</td>
</tr>
<tr>
<td></td>
<td>sw</td>
<td>$zero, 0($t2)</td>
</tr>
<tr>
<td></td>
<td>addi</td>
<td>$t0, $t0, 1</td>
</tr>
<tr>
<td></td>
<td>slt</td>
<td>$t3, $t0, $a1</td>
</tr>
<tr>
<td></td>
<td>bne</td>
<td>$t3, $zero, lp1</td>
</tr>
</tbody>
</table>

89
Comparison of Array vs. Pointer

- Multiply “strength reduced” to shift
- Array version requires shift to be inside loop
  - Part of index calculation for incremented i
  - c.f. incrementing pointer
- Compiler can achieve same effect as manual use of pointers
  - Induction variable elimination
  - Better to make program clearer and safer
### Summary: MIPS Operands

<table>
<thead>
<tr>
<th>Name</th>
<th>Example</th>
<th>Comments</th>
</tr>
</thead>
<tbody>
<tr>
<td>32 registers</td>
<td>$s0-$s7, $t0-$t9, $zero, $a0-$a3, $v0-$v1, $gp, $fp, $sp, $ra, $at</td>
<td>Fast locations for data. In MIPS, data must be in registers to perform arithmetic. MIPS register $zero always equals 0.</td>
</tr>
<tr>
<td>$2^{30}$ memory words</td>
<td>Memory[0], Memory[4], ..., Memory[4294967292]</td>
<td>Accessed only by data transfer instructions. MIPS uses byte addresses, so sequential words differ by 4. Memory holds data structures, such as arrays, and spilled registers, such as those saved on procedure calls.</td>
</tr>
</tbody>
</table>
## Summary: MIPS Assembly Language

<table>
<thead>
<tr>
<th>Category</th>
<th>Instruction</th>
<th>Example</th>
<th>Meaning</th>
<th>Comments</th>
</tr>
</thead>
<tbody>
<tr>
<td><strong>Arithmetic</strong></td>
<td>add</td>
<td>add $s1, $s2, $s3</td>
<td>$s1=$s2+$s3</td>
<td>Three register operands</td>
</tr>
<tr>
<td></td>
<td>subtract</td>
<td>sub $s1, $s2, $s3</td>
<td>$s1=$s2-$s3</td>
<td>Three register operands</td>
</tr>
<tr>
<td></td>
<td>add immediate</td>
<td>addi $s1, $s2, 100</td>
<td>$s1=$s2+100</td>
<td>Used to add constants</td>
</tr>
<tr>
<td><strong>Conditional</strong></td>
<td>branch on equal</td>
<td>beq $s1, $s2, 25</td>
<td>if ($s1==$s2) go to PC+4+100</td>
<td>Equal test; PC-relative branch</td>
</tr>
<tr>
<td></td>
<td>branch on not equal</td>
<td>bne $s1, $s2, 25</td>
<td>if ($s1!=$s2) go to PC+4+100</td>
<td>Not equal test; PC-relative</td>
</tr>
<tr>
<td></td>
<td>set on less than</td>
<td>slt $s1, $s2, $s3</td>
<td>if ($s2&lt;$s3) $s1=1 else $s1=0</td>
<td>Compare less than; for beq, bne</td>
</tr>
<tr>
<td></td>
<td>set less than immediate</td>
<td>slti $s1, $s2, 100</td>
<td>if ($s2&lt;100) $s1=1 else $s1=0</td>
<td>Compare less than constant</td>
</tr>
<tr>
<td><strong>Unconditional</strong></td>
<td>jump</td>
<td>j 2500</td>
<td>go to 10000</td>
<td>Jump to target address</td>
</tr>
<tr>
<td></td>
<td>jump register</td>
<td>jr $ra</td>
<td>go to $ra</td>
<td>For switch, procedure return</td>
</tr>
<tr>
<td></td>
<td>jump and link</td>
<td>jal 2500</td>
<td>$ra=PC+4; go to 10000</td>
<td>For procedure call</td>
</tr>
</tbody>
</table>
# Summary:
## MIPS Assembly Language

<table>
<thead>
<tr>
<th>Category</th>
<th>Instruction</th>
<th>Example</th>
<th>Meaning</th>
<th>Comments</th>
</tr>
</thead>
<tbody>
<tr>
<td><strong>Data transfer</strong></td>
<td>load word</td>
<td>lw $s1, 100($s2)</td>
<td>$s1=\text{Memory}[$s2+100]</td>
<td>Word from memory to register</td>
</tr>
<tr>
<td></td>
<td>store word</td>
<td>sw $s1, 100($s2)</td>
<td>\text{Memory}[$s2+100]=$s1</td>
<td>Word from register to memory</td>
</tr>
<tr>
<td></td>
<td>load half</td>
<td>lh $s1, 100($s2)</td>
<td>$s1=\text{Memory}[$s2+100]</td>
<td>Halfword from memory to register</td>
</tr>
<tr>
<td></td>
<td>store half</td>
<td>sh $s1, 100($s2)</td>
<td>\text{Memory}[$s2+100]=$s1</td>
<td>Halfword from register to memory</td>
</tr>
<tr>
<td></td>
<td>load byte</td>
<td>lb $s1, 100($s2)</td>
<td>$s1=\text{Memory}[$s2+100]</td>
<td>Byte from memory to register</td>
</tr>
<tr>
<td></td>
<td>store byte</td>
<td>sb $s1, 100($s2)</td>
<td>\text{Memory}[$s2+100]=$s1</td>
<td>Byte from register to memory</td>
</tr>
<tr>
<td></td>
<td>load upper immed.</td>
<td>lui $s1, 100</td>
<td>$s1=100*2^{16}</td>
<td>Loads constant in upper 16 bits</td>
</tr>
<tr>
<td><strong>Logical</strong></td>
<td>and</td>
<td>and $s1, $s2, $s3</td>
<td>$s1=$s2&amp;$s3</td>
<td>Bit-by-bit AND</td>
</tr>
<tr>
<td></td>
<td>or</td>
<td>or $s1, $s2, $s3</td>
<td>$s1=$s2</td>
<td>$s3</td>
</tr>
<tr>
<td></td>
<td>nor</td>
<td>nor $s1, $s2, $s3</td>
<td>$s1=\sim($s2</td>
<td>$s3)</td>
</tr>
<tr>
<td></td>
<td>and immediate</td>
<td>andi $s1, $s2, 100</td>
<td>$s1=$s2&amp;100</td>
<td>Bit-by-bit AND reg with constant</td>
</tr>
<tr>
<td></td>
<td>or immediate</td>
<td>ori $s1, $s2, 100</td>
<td>$s1=$s2</td>
<td>100</td>
</tr>
<tr>
<td></td>
<td>shift left logical</td>
<td>sll $s1, $s2, 10</td>
<td>$s1=$s2&lt;&lt;10</td>
<td>Shift left by constant</td>
</tr>
<tr>
<td></td>
<td>shift right logical</td>
<td>srl $s1, $s2, 10</td>
<td>$s1=$s2&gt;&gt;10</td>
<td>Shift right by constant</td>
</tr>
</tbody>
</table>